



Plataforma Inter-S Experimental (PIE)

Primeiros passos



Sumário

1. O que é o Inter-S?.....	3
2. Plataforma Inter-S Experimental.....	3
3. Exemplos de aplicação	6
3.1. Exemplo 1: Semáforo de LEDs.....	6
3.2. Exemplo 2: Teclado de massinha de modelar	10
3.3. Exemplo 3: Cancela eletrônica	14
3.4. Exemplo 4: Controle de um servo motor com potenciômetro	17
4. Esquema para montagem da PIE	21
5. Comandos INTER-S para operações com a plataforma PIE.....	24
5.1. Estado_USB	24
5.2. Mostra_Placa.....	24
5.3. Ative digitais / analógicas.....	25
5.4. Desative digitais / analógicas	25
5.5. Liga_Saída.....	25
5.6. Desliga_Saída	26
5.7. Servo_Motor_n / LServo_Motor_n.....	26
5.8. Analógica	26
5.9. Digital	27
5.10. P_Saídas	27



1. O que é o Inter-S?

Inter-S é um acrônimo do termo "Interpretador de Soluções", cujo significado expressa literalmente o seu objetivo, ou seja, é uma ferramenta que possibilita ao usuário escrever algoritmos lógicos (programas) que serão interpretados e executados pelo computador.

O Inter-S é um ambiente de programação dirigido aos iniciantes, desenvolvido especialmente para finalidades educacionais, servindo como ferramenta de apoio ao aprendizado da lógica e das técnicas de programação. O aplicativo utiliza uma linguagem simples, de fácil entendimento e com regras sintáticas claras baseadas na língua portuguesa, o que torna o desenvolvimento de programas uma tarefa descomplicada e intuitiva.

2. Plataforma Inter-S Experimental

A Plataforma Inter-S Experimental (PIE) é uma placa microcontrolada especialmente desenvolvida para o ambiente Inter-S. Essa plataforma, assim como os novos recursos de programação do Inter-S, têm como objetivo propiciar ao estudante ferramentas simples, tanto em nível de *hardware* como em nível de *software*, para o desenvolvimento de aplicações nas áreas de eletrônica e robótica.

A PIE é uma placa eletrônica que integra um microcontrolador PIC18F2550® (*Microchip Technology Inc.*), cuja conexão ao computador é feita através de uma porta USB utilizando o protocolo HID (*Human Interface Device*), não necessitando, portanto, de nenhum recurso adicional de *driver* para o seu reconhecimento pelo computador. A Figura 1, a seguir, apresenta o aspecto físico da plataforma PIE. (Na seção 4 deste documento você encontrará os detalhes técnicos da PIE, incluindo o seu diagrama esquemático, lista de componentes, *firmware* e outras informações necessárias para a sua confecção).





Figura 1 - Aspecto físico da Plataforma Inter-S Experimental

Para que você conheça os recursos da PIE, a Tabela 1 (apresentada a seguir) mostra uma breve descrição dos elementos que a integram. Nesta tabela, a localização de cada elemento está indicada por uma seta amarela.



Tabela 1: Descrição dos elementos da Plataforma Inter-S Experimental


Indicação do elemento (em amarelo)	Descrição
	Conector USB (tipo B - fêmea) Este componente permite a conexão da plataforma ao computador através de um cabo USB.
	Botão "Reset" Este botão reinicia a plataforma, estabelecendo nova conexão com o computador.
	Saídas digitais A plataforma disponibiliza oito saídas digitais identificadas pelos números de 0 a 7. Essas saídas são acionadas por meio de programa e podem assumir nível lógico "um" (ligada) ou nível lógico "zero" (desligada). Para cada saída existe um led acoplado que indica o estado da mesma (led aceso = ligada / led apagado = desligada). Existe também uma barra de terminais, onde cada terminal está ligado a uma saída. Esses terminais podem ser utilizados para ligações externas (para acionamento de leds, relés, motores, etc.).
	Saídas para Servos Motores As quatro primeiras saídas digitais (saídas 0 a 3) também podem ser utilizadas para o controle de servos-motores, os quais são identificados pelos números 1, 2, 3 e 4 e estão associados às saídas 0, 1, 2 e 3, respectivamente.
	Entradas analógicas Existem três entradas analógicas que podem ser utilizadas para a leitura de potenciômetros ou sensores diversos. Cada entrada possui conversor analógico-digital de 8 bits, podendo gerar um valor entre 0 e 255, de acordo com a tensão de entrada, a qual pode variar entre 0 e 5 volts. Para facilitar a conexão de sensores e principalmente potenciômetros, cada entrada analógica contém um pino que fornece 5 volts, um pino terra (GND), além do pino específico da entrada (pino central).



Entradas digitais


São utilizadas para a conexão de chaves, botões ou sensores digitais externos (componentes que podem assumir apenas dois estados: ligado ou desligado).

Essas entradas são identificadas pelos números de 1 a 4 e podem ser acessadas pelos terminais correspondentes, os quais estão dispostos em uma única barra de terminais.



Botões


A plataforma experimental possui quatro botões táteis dispostos em forma de "joystick", os quais estão acoplados às entradas digitais 1, 2, 3 e 4. Esses botões são ligados às entradas digitais por meio de uma montagem do tipo *pull-up*, ou seja, quando pressionados eles fornecem valor lógico zero (0 volt) à entrada correspondente, e quando soltos, fornecem valor lógico um (5 volts).



Saída de tensão


Esta saída fornece tensão de 5 volts extraída diretamente da porta USB do computador, podendo ser utilizada para a alimentação de pequenos dispositivos externos que não exijam correntes altas.

Existem três terminais que fornecem 5 volts cada e três terminais terra (GND).



Jumper LEDs (J1)

Este *jumper* ativa os leds que estão conectados às saídas digitais. Se for preciso desativar esses leds, basta remover o *jumper* J1.



Jumper Botões (J2)

Este *jumper* ativa os botões táteis que estão conectados às entradas digitais. Se for preciso desativar esses botões, basta remover o *jumper* J2.



3. Exemplos de aplicação

Nesta seção serão apresentados alguns exemplos que ilustram a utilização da Plataforma Inter-S Experimental.

3.1. Exemplo 1: Semáforo de LEDs

Neste exemplo serão demonstrados os procedimentos para a elaboração de um simples semáforo construído com três LEDs (*Light-Emitting Diode* - Diodo emissor de luz), cuja montagem é apresentada na Figura 2.




Figura 2 - Montagem do semáforo de LEDs

Materiais necessários:

- Plataforma Inter-S Experimental;
- Matriz de contatos (*protoboard*);
- 3 resistores de 470Ω;
- 3 leds (vermelho, amarelo e verde);
- Fios para ligações.

Nota: Para testar o programa proposto não é necessária a realização da montagem mostrada na Figura 2. Com apenas a Plataforma Experimental é possível realizar o projeto, pois, o funcionamento do semáforo pode ser observado por meio dos LEDs existentes na própria plataforma (LEDs acoplados às saídas digitais). Também é possível executar este programa mesmo que você não tenha a plataforma PIE, neste caso, o resultado do processamento pode ser observado na imagem virtual da plataforma, que é exibida ao lado da janela de execução do Inter-S.



Programa

Usando o Inter-S, devemos elaborar um programa para realizar os seguintes passos:


Passo 1: Acender o LED vermelho e apagá-lo após 5 segundos

Passo 2: Acender o LED verde e apagá-lo após 3 segundos

Passo 3: Acender o LED amarelo e apagá-lo após 2 segundos

Passo 4: Repetir os procedimentos a partir do passo 1

A Figura 3 a seguir mostra o programa proposto após sua digitação na janela de edição do Inter-S.



```
Inter-S Versão 3.0 (_PIE_semaforo_simples.ALG)
Arquivo  Editar  Executar  Diversos  Sair
Novo  Abrir  Salvar  Imprimir  Indentar  Executar  Tabela de cores  Ajuda  Porta serial  Baud Rates

Rotina
  Declare tecla Numérico
  Mostra_Placa
  Repita
    tecla=Ftecla
    Liga_Saída 0; Atrase 5000; Desliga_Saída 0
    Liga_Saída 2; Atrase 3000; Desliga_Saída 2
    Liga_Saída 1; Atrase 2000; Desliga_Saída 1
  AtéQue tecla=27
FimRotina
```

Figura 3 - Programa digitado na janela de edição do Inter-S

Comentários sobre o programa

Na primeira linha temos a instrução "Rotina", que é obrigatória para determinar o início do programa. Em seguida, por meio da instrução "Declare", devemos declarar as variáveis que serão utilizadas na rotina (neste caso usaremos apenas a variável "tecla", que será responsável pelo recebimento do código da tecla digitada pelo usuário durante a execução do programa).

Na terceira linha da rotina temos a instrução "mostra_placa", que exibe na tela uma imagem virtual da Plataforma Inter-S Experimental, na qual é possível verificar, em tempo real, o estado da plataforma conectada ao computador (este recurso pode ser usado mesmo que a plataforma não esteja conectada, permitindo dessa forma a realização de simulações).

A instrução "Repita" inicia uma estrutura de repetição, na qual as instruções entre "Repita" e "AtéQue tecla=27" serão executadas repetidamente até que a variável "tecla" tenha conteúdo igual a 27. O código 27 corresponde à tecla "ESC", dessa forma, a estrutura de repetição será encerrada quando o usuário pressionar "ESC".

Na quinta linha da rotina temos a instrução "tecla=Ftecla". Esta instrução é responsável por fazer uma varredura no teclado do computador e atribuir à variável "tecla" o código correspondente à tecla pressionada (caso alguma tecla esteja pressionada nesse momento). Caso nenhuma tecla tenha sido pressionada, a variável "tecla" receberá valor 0 (zero).

A sexta linha da rotina contém três instruções, sendo elas: "Liga_Saída 0" (liga a saída digital número 0, acendendo o LED vermelho), "Atrase 5000" (gera uma pausa de 5000 milissegundos - 5 segundos) e "Desliga_Saída 0" (desliga a saída digital número 0, apagando o LED vermelho). Observe que essas instruções realizam o passo 1 da nossa rotina: "Acender o LED vermelho e apagá-lo após 5 segundos".

Da mesma forma, a sétima linha contém três instruções que realizam o passo 2 da nossa rotina: "Acender o LED verde e apagá-lo após 3 segundos". Com a mesma ideia, a oitava linha realiza o passo 3: "Acender o LED amarelo e apagá-lo após 2 segundos".

A nona linha da rotina, que contém a instrução "AtéQue tecla=27", determina a condição de finalização da estrutura de repetição. Como citado anteriormente, a estrutura de repetição será



finalizada quando o valor da variável "tecla" for igual a 27, o que indica que o usuário pressionou a tecla "ESC".

A décima e última linha da rotina, que contém a instrução "FimRotina", finaliza o programa.

Melhorias no programa

Se você construiu e executou o programa proposto, deve ter percebido que a janela de execução do Inter-S permaneceu em branco durante todo o processamento (somente a imagem virtual da Plataforma Experimental apresentou animação). Para construir uma interface gráfica que nos mostre o funcionamento do semáforo em tempo real, podemos utilizar alguns recursos do Inter-S para gerar uma animação durante o processamento. O programa proposto a seguir, cujo código é mostrado na Figura 4, exibirá na janela de execução do Inter-S um semáforo virtual que funcionará em conformidade com a Plataforma Experimental.

```
Inter-S Versão 3.0 (PIE_semaforo.ALG)
Arquivo Editar Executar Diversos Sair
Novo Abrir Salvar Imprimir Indentar Executar Tabela de cores Ajuda Porta serial Baud Rates Linha: 27

Rotina
Declare tecla Numérico
Mostra_Placa
Moldura 5,35,9,45,branco
Moldura 11,35,15,45,branco
Moldura 17,35,21,45,branco
Repita
  tecla=Ftecla
  Liga_Saída 0
  Moldura 5,35,9,45,vermelho
  Atrase 5000
  Moldura 5,35,9,45,branco
  Desliga_Saída 0
  Liga_Saída 2
  Moldura 17,35,21,45,verde+
  Atrase 3000
  Moldura 17,35,21,45,branco
  Desliga_Saída 2
  Liga_Saída 1
  Moldura 11,35,15,45,amarelo
  Atrase 2000
  Moldura 11,35,15,45,branco
  Desliga_Saída 1
  AtéQue tecla=27
FimRotina
```

Figura 4 - Programa "semáforo" com interface gráfica

Comentários sobre o novo programa

A tabela a seguir mostra as instruções comentadas do novo programa.

Instruções	Comentários
Rotina	Início da rotina
Declare tecla NUMÉRICO	Declaração da variável utilizada (tecla)
Mostra_Placa	Exibe imagem virtual da plataforma experimental
Moldura 5,35,9,45,branca	Exibe na tela uma moldura branca, da linha 5 e coluna 35 até a linha 9 e coluna 45, simbolizando o sinal vermelho apagado.
Moldura 11,35,15,45,branca	Exibe na tela uma moldura branca, da linha 11 e coluna 35 até a linha 15 e coluna 45, simbolizando o sinal amarelo apagado.
Moldura 17,35,21,45,branca	Exibe na tela uma moldura branca, da linha 17 e coluna 35 até a linha 21 e coluna 45, simbolizando o sinal verde apagado.
Repita	Inicia a estrutura de repetição principal
tecla=Ftecla	Verifica se alguma tecla foi pressionada e coloca o valor da mesma na variável "tecla"



Liga_Saída 0	Liga a saída 0 (acende o LED vermelho)
Moldura 5,35,9,45,vermelho	Exibe na tela uma moldura com preenchimento vermelho, da linha 5 e coluna 35 até a linha 9 e coluna 45, indicando sinal "PARE" aceso.
Atrase 5000	Gera uma pausa de 5000 ms (5 segundos)
Moldura 5,35,9,45,branco	Exibe na tela uma moldura branca, da linha 5 e coluna 35 até a linha 9 e coluna 45, indicando sinal vermelho apagado.
Desliga_Saída 0	Desliga a saída 0 (apaga o LED vermelho)
Liga_Saída 2	Liga a saída 2 (acende o LED verde)
Moldura 17,35,21,45,verde+	Exibe na tela uma moldura com preenchimento verde, da linha 17 e coluna 35 até a linha 21 e coluna 45, indicando sinal "SIGA" aceso.
Atrase 3000	Gera uma pausa de 3000 ms (3 segundos)
Moldura 17,35,21,45,branco	Exibe na tela uma moldura branca, da linha 17 e coluna 35 até a linha 21 e coluna 45, indicando sinal verde apagado.
Desliga_Saída 2	Desliga a saída 2 (apaga o LED verde)
Liga_Saída 1	Liga a saída 1 (acende o LED amarelo)
Moldura 11,35,15,45,amarelo	Exibe na tela uma moldura com preenchimento amarelo, da linha 11 e coluna 35 até a linha 15 e coluna 45, indicando sinal "ATENÇÃO" aceso.
Atrase 2000	Gera uma pausa de 2000 ms (2 segundos)
Moldura 11,35,15,45,branco	Exibe na tela uma moldura com preenchimento branco, da linha 11 e coluna 35 até a linha 15 e coluna 45 para indicar sinal amarelo apagado.
Desliga_Saída 1	Desliga a saída 1 (apaga o LED amarelo)
AtéQue tecla=27	Se a variável "tecla" for igual a 27 (o que indica que o usuário pressionou a tecla "ESC") o programa será encerrado, caso contrário, as instruções dentro da estrutura de repetição principal serão executadas novamente.
FimRotina	Fim da rotina



3.2. Exemplo 2: Teclado de massinha de modelar

Neste exemplo serão demonstrados os procedimentos para a construção de um pequeno teclado de três teclas confeccionadas com massinha de modelar, cuja montagem é apresentada na Figura 5.




Figura 5 - Montagem do teclado

Materiais necessários:

- Plataforma Inter-S Experimental;
- Matriz de contatos (*protoboard*);
- 3 resistores de 330KΩ;
- Massinha de modelar;
- Fios para ligações.

Nota: As teclas de massinha de modelar devem ser conectadas ao circuito por meio de cabos tipo "jumper" ou fios com extremidades descascadas, conforme mostra a Figura 6.




Figura 6 - Forma de conexão das teclas de massinha com o circuito



Programa

Para testar o teclado, o programa Inter-S deve realizar os seguintes passos:

- Passo 1: Exibir na tela 3 molduras com preenchimento verde escuro (estas molduras indicarão o acionamento das teclas de massinha, ou seja, quando o usuário tocar em uma tecla de massinha, a moldura correspondente a ela será preenchida com a cor verde claro);
- Passo 2: Ler as entradas analógicas e guardar os valores lidos;
- Passo 3: Verificar se o valor lido de cada entrada analógica é menor que 220, o que significa que o usuário tocou na tecla de massinha correspondente àquela entrada. Neste caso, o programa deve alterar a cor de preenchimento da moldura para verde claro.

Breve explicação sobre as entradas analógicas

As entradas analógicas da plataforma PIE possuem conversores analógicos-digitais de 8 bits, isso significa que os valores apresentados por essas entradas podem variar de 0 a 255. Por exemplo, se alimentarmos uma entrada analógica com 5 volts (máximo), o valor apresentado por ela será 255; se alimentarmos com 2,5 volts, o valor apresentado será 128 (aproximadamente a metade de 255); se alimentarmos com 0 volt, o valor apresentado será 0 (zero). Matematicamente, podemos então definir o valor apresentado por uma entrada analógica por meio da seguinte equação:

$$X = \frac{255 \times V_i}{5}$$

onde:

X = Valor apresentado pela entrada analógica

V_i = Tensão aplicada na entrada (em volts)

Por exemplo, se aplicarmos 1 volt em uma determinada entrada analógica, teremos como resposta o valor 51:

$$X = \frac{255 \times 1}{5} = 51$$

Conforme montagem apresentada na Figura 5, as entradas analógicas 1, 2 e 3 são alimentadas com 5V, isso faz com que os seus valores cheguem ao máximo (255). Quando o usuário tocar em uma das teclas de massinha, segurando ao mesmo tempo o fio que está conectado ao GND (terra), ocorrerá um fechamento de circuito entre a entrada analógica e o terminal GND, o qual apresenta 0V (zero volt), desta forma, a tensão de entrada tenderá a zero, fazendo com que o valor apresentado pela entrada analógica seja também zero. Em resumo, quando o usuário tocar em uma das teclas de massinha, o valor apresentado pela entrada analógica será próximo de zero. Por outro lado, quando a tecla não for tocada, o valor apresentado será próximo de 255.

A Figura 7 a seguir mostra o programa já digitado na janela de edição do Inter-S.



```
Inter-S Versão 3.0 (_PIE_massinha.ALG3)
Arquivo  Editar  Executar  Diversos  Sair
Novo  Abrir  Salvar  Imprimir  Indentar  Executar  Tabela de cores  Ajuda  Porta serial  Baud Rates  Linha: 2

Rotina
  Declare a1,a2,a3,tecla Numéricos
  Mostra_Placa
  Ative analógicas
  Moldura 12,16,17,26,verde
  Moldura 12,36,17,46,verde
  Moldura 12,56,17,66,verde
  Repita
    tecla=Ftecla
    a1=analógica 1; a2=analógica 2; a3=analógica 3
    Se a1 < 220 Então
      Moldura 12,16,17,26,lima
      Liga_Saída 0
    Senão
      Desliga_Saída 0
      Moldura 12,16,17,26,verde
    FimSe
    Se a2 < 220 Então
      Moldura 12,36,17,46,lima
      Liga_Saída 1
    Senão
      Desliga_Saída 1
      Moldura 12,36,17,46,verde
    FimSe
    Se a3 < 220 Então
      Moldura 12,56,17,66,lima
      Liga_Saída 2
    Senão
      Desliga_Saída 2
      Moldura 12,56,17,66,verde
    FimSe
  AtéQue tecla=27
FimRotina
```

Figura 7 - Programa proposto

Comentários sobre o programa

Instruções	Comentários
Rotina	Início da rotina
Declare a1,a2,a3,tecla NUMÉRICO	Declaração das variáveis utilizadas (a1, a2 e a3 serão usadas para guardar os valores lidos das entradas analógicas. A variável tecla será usada para encerrar o programa mediante opção do usuário)
mostra_placa	Exibe imagem virtual da plataforma experimental
Ative analógicas	Ativa a leitura das entradas analógicas
Moldura 12,16,17,26,verde	Exibe na tela uma moldura com preenchimento verde escuro, correspondente à 1ª tecla de massinha.
Moldura 12,36,17,46,verde	Exibe na tela uma moldura com preenchimento verde escuro, correspondente à 2ª tecla de massinha.
Moldura 12,56,17,66,verde	Exibe na tela uma moldura com preenchimento verde escuro, correspondente à 3ª tecla de massinha.
Repita	Inicia a estrutura de repetição principal.
tecla=Ftecla	Verifica se alguma tecla foi pressionada (no teclado do computador) e coloca o valor da mesma na variável "tecla".
a1=analógica 1;a2=analógica 2;a3=analógica 3	Lê as entradas analógicas 1, 2 e 3 e armazena os valores lidos nas variáveis a1, a2 e a3.
Se a1 < 220 então	Verifica se o valor de a1 é menor que 220, o que significa que a tecla de massinha da entrada analógica 1 foi tocada. Caso positivo...



Moldura 12,16,17,26,lima	... muda a cor da moldura para verde claro (lima)
Liga_Saída 0	... liga a saída 0 (acende o led 0 da PIE)
Senão	Caso contrário (se o valor não for menor que 220)...
Desliga_Saída 0	... desliga a saída 0 (apaga o led 0 da PIE)
Moldura 12,16,17,26,verde	... muda a cor da moldura para verde escuro
FimSe	Finaliza a estrutura de decisão
Se a2 < 220 então	Verifica se o valor de a2 é menor que 220, o que significa que a tecla de massinha da entrada analógica 2 foi tocada. Caso positivo...
Moldura 12,36,17,46,lima	... muda a cor da moldura para verde claro (lima)
Liga_Saída 1	... liga a saída 1 (acende o led 1 da PIE)
Senão	Caso contrário (se o valor não for menor que 220)...
Desliga_Saída 1	... desliga a saída 1 (apaga o led 1 da PIE)
Moldura 12,36,17,46,verde	... muda a cor da moldura para verde escuro
FimSe	Finaliza a estrutura de decisão
Se a3 < 220 então	Verifica se o valor de a3 é menor que 220, o que significa que a tecla de massinha da entrada analógica 3 foi tocada. Caso positivo...
Moldura 12,56,17,66,lima	... muda a cor da moldura para verde claro (lima)
Liga_Saída 2	... liga a saída 2 (acende o led 2 da PIE)
Senão	Caso contrário (se o valor não for menor que 220)...
Desliga_Saída 2	... desliga a saída 2 (apaga o led 2 da PIE)
Moldura 12,56,17,66,verde	... muda a cor da moldura para verde escuro
FimSe	Finaliza a estrutura de decisão
AtéQue tecla=27	Se a variável "tecla" for igual a 27 (o que indica que o usuário pressionou a tecla "ESC" do teclado do computador) o programa será encerrado, caso contrário, as instruções dentro da estrutura de repetição principal serão executadas novamente.
FimRotina	Fim da rotina

3.3. Exemplo 3: Cancela eletrônica

Neste exemplo, iremos construir um dispositivo para acionar uma cancela eletrônica montada sobre um servo motor. O objetivo do dispositivo é promover a abertura da cancela quando for detectada a presença de um veículo e, após a passagem desse veículo, promover o fechamento da cancela. Para a detecção do veículo, usaremos um LDR (*Light Dependent Resistor* - Resistor Dependente de Luz) acoplado à entrada analógica 3, conforme montagem apresentada na Figura 8.




Figura 8 - Montagem do circuito da cancela eletrônica

Materiais necessários:

- Plataforma Inter-S Experimental;
- Matriz de contatos (*protoboard*);
- 1 resistor de 1KΩ;
- 1 LDR (resistor dependente de luz)
- 1 servo-motor pequeno;
- Fios para ligações.

Programa

Usando o Inter-S, devemos escrever um programa para realizar as seguintes ações:



- Verificar se o usuário pressionou a tecla "ESC". Caso positivo, o programa deve ser encerrado;
- Se a tecla "ESC" não foi pressionada, o programa deve ler o valor da entrada analógica 3, na qual está acoplado o LDR;
- Se o valor da entrada analógica 3 for menor que 120, isso significa que a incidência de luz sobre o LDR é pequena, o que indica a presença de um "veículo" em frente ao sensor. Neste caso, o servo-motor deve ser acionado para que a cancela se abra;
- Após a abertura da cancela, o programa deve aguardar até que a entrada analógica 3 tenha um valor superior a 120, o que indica que o "veículo" não se encontra mais em frente ao sensor. Neste caso, o programa deve aguardar 1 segundo e em seguida acionar o servo-motor para fechar a cancela. Após o fechamento, repete-se o processo até que a tecla "ESC" seja pressionada, o que causará o encerramento do programa.

A Figura 9 a seguir mostra o programa já digitado na janela de edição do Inter-S.

```
Inter-S Versão 3.0 (C:\Users\Roberto\Desktop\EXEMPLOS_INTERS\_PIE_cancela.ALG3)
Arquivo  Editar  Executar  Diversos  Sair
Novo  Abrir  Salvar  Imprimir  Indentar  Executar  Tabela de cores  Ajuda  Porta serial  Baud Rates  Linha: 31

Rotina
  Declare a1,tecla,estado Numéricos
  Mostra_Placa
  Active analógicas
  LServo_Motor_1 13
  Escreva "CANCELA FECHADA"
  estado=0
  [inicio]
  tecla=Ftecla
  Se tecla = 27 Então
    Interrompa
  FimSe
  a1=analógica 3
  Se a1 < 120 e estado=0 Então
    Limpa
    Escreva "CANCELA ABERTA"
    LServo_Motor_1 55
    estado=1
  Senão
    Se a1 >= 120 e estado=1 Então
      Atrase 1000
      Limpa
      Escreva "CANCELA FECHADA"
      LServo_Motor_1 13
      estado=0
    FimSe
  FimSe
  VaPara inicio
FimRotina
```

Figura 9 - Programa proposto

Comentários sobre o programa

Instruções	Comentários
Rotina	Início da rotina
Declare a1,tecla,estado Numéricos	Declaração das variáveis utilizadas. A variável "a1" será usada para guardar o valor lido da entrada analógica 3; a variável "tecla" será usada para encerrar o programa mediante opção do usuário e a variável "estado" indicará se a cancela está aberta (0) ou fechada (1).
Mostra_Placa	Exibe imagem virtual da plataforma experimental



Ative analógicas	Ativa a leitura das entradas analógicas.
LServo_Motor_1 13	Aciona o servo-motor 1 levando-o para o ângulo de 13 graus (este ângulo corresponde à posição da cancela fechada - você deve alterar esse ângulo de acordo com a sua montagem).
Escreva "CANCELA FECHADA"	escreve na tela a expressão "CANCELA FECHADA"
estado=0	Atribui valor 0 (zero) à variável "estado" (isso indica que a cancela está fechada).
[inicio]	Este é um rótulo que indica um ponto de retorno do programa. Não pode ser acentuado e deve ser escrito entre colchetes.
tecla=Ftecla	Verifica se alguma tecla foi pressionada (no teclado do computador) e coloca o valor da mesma na variável "tecla".
Se tecla = 27 Então	Inicia uma estrutura de decisão que verifica se o valor da variável "tecla" é igual a 27 (que corresponde à tecla "ESC"). Caso positivo...
Interrompa	... encerra o programa.
FimSe	Finaliza a estrutura de decisão.
a1=analógica 3	Lê a entrada analógica 3 e guarda o valor lido na variável "a1".
Se a1 < 120 e estado=0 Então	Inicia uma estrutura de decisão que verifica se o valor de "a1" é menor que 120 e se o valor da variável "estado" é igual a zero, o que significa que existe um "veículo" em frente ao sensor e a cancela está fechada. Caso o resultado da verificação for verdadeiro...
Limpa	... limpa a tela, ...
Escreva "CANCELA ABERTA"	... escreve na tela a expressão "CANCELA ABERTA", ...
LServo_Motor_1 55	... aciona o servo-motor 1 levando-o para o ângulo de 55 graus (abre a cancela) e ...
estado=1	... atribui o valor 1 (um) à variável "estado", o que indica que a cancela está aberta.
Senão	Caso contrário (se "a1" não for menor que 120 ou "estado" for diferente de zero)...
Se a1 >= 120 e estado=1 Então	Inicia outra estrutura de decisão que verifica se o valor de "a1" é maior ou igual a 120 e se o valor da variável "estado" é igual a 1, o que significa que não existe "veículo" em frente ao sensor e a cancela está aberta. Caso o resultado da verificação for verdadeiro...
Atrase 1000	... gera uma pausa de 1000ms (1 segundo), ...
Limpa	... limpa a tela, ...
Escreva "CANCELA FECHADA"	... escreve na tela a expressão "CANCELA FECHADA", ...
LServo_Motor_1 13	... aciona o servo-motor 1 levando-o para o ângulo de 13 graus (fecha a cancela).
estado=0	... atribui o valor 0 (zero) à variável "estado", o que indica que a cancela está fechada.
FimSe	Finaliza a estrutura de decisão "Se a1 >= 120 e estado=1"
FimSe	Finaliza a estrutura de decisão "Se a1 < 120 e estado=0"
VaPara inicio	Desvia o programa para o rótulo [inicio], gerando uma estrutura de repetição que só será finalizada quando o usuário pressionar a tecla "ESC".
FimRotina	Fim da rotina

3.4. Exemplo 4: Controle de um servo motor com potenciômetro

Neste exemplo iremos construir um pequeno dispositivo para controlar um servo motor por meio de um potenciômetro acoplado à entrada analógica 1 da plataforma experimental. O giro do servo motor deverá acompanhar o potenciômetro, isto é, quando o usuário girar o eixo do potenciômetro, o servo motor deverá também girar na mesma proporção e direção.

Usaremos um potenciômetro de 1K Ω , acoplando o seu terminal da esquerda no conector 5V, o seu terminal da direita no conector GND (0V) e o seu terminal central na entrada analógica 1. Dessa forma, quando o eixo do potenciômetro for girado totalmente para a esquerda, serão enviados 5 volts para a entrada analógica (valor máximo). Por outro lado, quando o eixo for girado totalmente para a direita, será enviado zero volt para a entrada analógica (valor mínimo).

Sabendo-se que as entradas analógicas da plataforma experimental PIE são dotadas de conversores A/D (analógicos/digitais) de 8 bits, teremos nas saídas desses conversores um valor entre 0 (zero) e 255 (duzentos e cinquenta e cinco). Dessa forma, quando o potenciômetro estiver com seu eixo totalmente voltado para a esquerda, a entrada analógica receberá 5 volts, apresentando um valor de 255 (máximo), e quando o seu eixo estiver totalmente voltado para a direita, a entrada analógica receberá zero volt, apresentando valor zero (mínimo). Quando o eixo estiver em uma posição qualquer, o valor recebido pela entrada analógica será uma fração do valor máximo de 5 volts, proporcional à posição do eixo.

O motor que iremos usar é um servo de 180°, que permite um deslocamento angular dentro da faixa de 0° a 180° apenas. Devemos então considerar que quando o valor apresentado pela entrada analógica for 255 (máximo), o ângulo do servo motor deve ser 180°, e quando o valor apresentado for zero (mínimo), o ângulo do servo deve ser 0°. Dessa forma, podemos usar uma simples regra de três para encontrar o ângulo certo em função do valor apresentado pela entrada analógica. Matematicamente, teremos:

$$\text{Ângulo} = \frac{180 \times \text{Valor da entrada analógica } 1}{255}$$

A Figura 10 apresenta a montagem do circuito proposto.




Figura 10 - Montagem do dispositivo



Materiais necessários:

- Plataforma Inter-S Experimental;
- 1 potenciômetro linear de 1K Ω ;
- 1 servo-motor pequeno;
- Fios para ligações.

Programa

Usando o Inter-S, devemos escrever um programa para realizar os seguintes passos:

1. Verificar se o usuário pressionou a tecla "ESC". Caso positivo, o programa deve ser encerrado;
2. Se a tecla "ESC" não foi pressionada, o programa deve ler o valor da entrada analógica 1, na qual está acoplado o potenciômetro;
3. Usar a regra de três para encontrar o ângulo em função do valor apresentado pela entrada analógica 1;
4. Enviar o ângulo encontrado para o servo motor;
5. Exibir na tela de execução uma imagem representando o servo motor e o ângulo no qual está posicionado (animação de tela);
6. Repetir os procedimentos a partir do passo 1.

A Figura 11 mostra o programa já digitado na janela de edição do Inter-S.

```
Inter-S Versão 3.0 (C:\Users\Roberto\Desktop\EXEMPLOS_INTERS\_PIE_servo.ALG3)
Arquivo  Editar  Executar  Diversos  Sair
Novo  Abrir  Salvar  Imprimir  Indentar  Executar  Tabela de cores  Ajuda  Porta serial  Baud Rates  Linha: 38

Rotina
Declare pi, x, y, angulo, raio Numéricos
Declare tecla, a1 Numéricos
Declare radianos Numéricos
pi=3.14159
raio=25
Mostra_Placa
Ative analógicas
Escala 10
TamanhoPonto 8
CorPonto ouro
CorFundo branco
[volta]
tecla=Ftecla
Se tecla=27 Então
    Limpa
    Interrompa
FimSe
a1=analógica 1
angulo=(180*a1)/255
Servo_Motor_1 angulo
Limpa
Círculo 0,0,10,celeste
radianos = angulo*(pi/180)
x=cos(radianos)*raio
y=seno(radianos)*raio
Linha 0,0/x,y
Posicione 19,35
Escreva "Ângulo: ",trunca(angulo),"°"
VaPara volta
FimRotina
```

Figura 11 -Programa proposto (controle de servo motor com potenciômetro)



Comentários sobre o programa

Instruções	Comentários
Rotina	Início da rotina
<code>Declare pi,x,y,angulo,raio Numéricos</code>	Declaração das variáveis utilizadas. A variável "pi" armazenará o valor do PI radiano (π); as variáveis "x" e "y" armazenarão a coordenada gráfica para indicar na tela o ângulo atual do servo motor; a variável "angulo" armazenará o ângulo para posicionamento do servo motor; a variável "raio" armazenará o comprimento da haste que representará o eixo do servo motor na tela.
<code>Declare tecla, a1 Numéricos</code>	Declaração das variáveis utilizadas. A variável "tecla" armazenará o código da tecla digitada pelo usuário; a variável "a1" armazenará o valor fornecido pela entrada analógica 1.
<code>Declare radianos Numéricos</code>	Declaração da variável "radianos", que será usada para converter o ângulo de graus para radianos. Essa conversão será necessária porque as funções trigonométricas "seno" e "cosseno" do Inter-S trabalham com argumentos em radianos.
<code>pi=3.14159</code>	Atribui à variável "pi" o valor do PI radiano (π)
<code>raio=25</code>	Atribui à variável "raio" o valor 25 (comprimento, em pontos gráficos, da haste que será exibida na tela para representar o eixo do servo motor)
<code>Mostra_Placa</code>	Exibe imagem virtual da plataforma experimental
<code>Active analógicas</code>	Ativa a leitura das entradas analógicas.
<code>Escala 10</code>	Determina o número de pontos gráficos da grade da área gráfica. O valor 10 (dez) determina que o espaço entre as linhas que formam a grade conterá 10 pontos gráficos tanto na horizontal como na vertical. A área gráfica é tratada como um plano cartesiano, onde a origem (ponto 0,0) fica no centro da janela de execução. Consulte o módulo de ajuda do Inter-S para maiores detalhes sobre a área gráfica.
<code>TamanhoPonto 8</code>	Define o tamanho do ponto gráfico (número de pixels).
<code>CorPonto ouro</code>	Define a cor dos pontos gráficos (ouro).
<code>CorFundo branco</code>	Define a cor do fundo da janela gráfica (branco).
<code>[volta]</code>	Este é um rótulo que indica um ponto de retorno do programa.
<code>tecla=Ftecla</code>	Verifica se alguma tecla foi pressionada (no teclado do computador) e coloca o valor da mesma na variável "tecla".
<code>Se tecla = 27 Então</code>	Inicia uma estrutura de decisão que verifica se o valor da variável "tecla" é igual a 27 (que corresponde à tecla "ESC"). Caso positivo...
<code>Limpa</code>	... limpa (apaga) a tela e ...
<code>Interrompa</code>	... encerra o programa.
<code>FimSe</code>	Finaliza a estrutura de decisão.
<code>a1=analógica 1</code>	Lê a entrada analógica 1 e guarda o valor lido na variável "a1".
<code>angulo=(180*a1)/255</code>	Aplica a regra de três para calcular o ângulo que deve ser enviado ao servo motor, com base no valor obtido na entrada analógica 1. Observe que a variável "angulo" receberá o resultado do cálculo $(180*a1)/255$.
<code>Servo_Motor_1 angulo</code>	Envia para o servo motor o ângulo calculado. Esta instrução fará com que o servo motor se posicione no ângulo informado.
<code>Limpa</code>	Limpa a tela.
<code>Círculo 0,0,10,celeste</code>	Desenha um círculo preenchido com cor celeste, com centro na coordenada 0,0 (centro da tela) e com raio de 10 pontos.
<code>radianos = angulo*(pi/180)</code>	Atribui à variável "radianos" o valor do ângulo convertido em radianos.
<code>x=cos(radianos)*raio</code>	Calcula o valor da variável "x", que representa a abscissa (coordenada X) do ponto correspondente ao ângulo para apresentação na tela (somente para efeito de animação).



<code>y=seno(radianos)*raio</code>	Calcula o valor da variável "y", que representa a ordenada (coordenada Y) do ponto correspondente ao ângulo para apresentação na tela (somente para efeito de animação).
<code>Linha 0,0/x,y</code>	Traça uma linha do centro da tela (coordenada 0,0) até a coordenada x,y (somente para efeito de animação). Os valores de "x" e "y" foram calculados nas instruções anteriores e representam a posição (ângulo) em que se encontra o servo motor.
<code>Posicione 19,35</code>	Posiciona o cursor na linha 19 e coluna 35.
<code>Escreva "Ângulo: ",trunca(angulo),"°"</code>	Escreve na tela, a partir da posição do cursor, a expressão "Ângulo: " seguida do valor truncado do ângulo (valor sem casas decimais) e do símbolo de grau (°). Dessa forma, além da animação de tela que representa a movimentação do servo motor, o usuário terá a indicação do ângulo atual.
<code>VaPara volta</code>	Desvia o programa para o rótulo [volta], gerando uma estrutura de repetição que só será finalizada quando o usuário pressionar a tecla "ESC".
<code>FimRotina</code>	Fim da rotina



4. Esquema para montagem da PIE

O circuito da Plataforma Inter-S Experimental (PIE) segue o modelo clássico recomendado pelo próprio fabricante do microcontrolador PIC (*Microchip Technology Inc.*), com a adição de alguns componentes de interfaceamento, como botões, conectores e leds de sinalização, que permitem ao usuário realizar suas montagens experimentais com maior facilidade. A Figura 12 mostra o diagrama esquemático da plataforma.





Figura 12 - Diagrama esquemático da Plataforma Inter-S Experimental

Lista de materiais


- 1 microcontrolador PIC18F2550 I/ISP (DIP28 slim)
- 1 soquete DIP28 slim estampado
- 9 resistores de 200Ω (R1 a R9)
- 5 resistores de 10KΩ (R10 a R14)
- 2 capacitores cerâmicos de 22pF (C2 e C3)
- 2 capacitores cerâmicos de 100nF (C1 e C4)
- 1 cristal oscilador de 20MHz, caneca baixa (X1)
- 4 leds retangulares de 3mm, verdes (D1 a D4)
- 4 leds retangulares de 3mm, amarelos (D5 a D8)
- 1 led retangular de 3mm, vermelho (D9)
- 5 botões táteis, 8x8mm, 4 terminais (B1 a B4 e RESET)
- 1 conector USB, tipo B, fêmea, para soldagem em placa (CON1)
- Barra de terminais 11,2mm macho e 2 jumpers (J1 e J2)
- Barra de pinos 11,2mm fêmea (saídas digitais, entradas digitais, entradas analógicas, GND e 5V)




Imagens e máscaras para a confecção da PCI




Aspecto da placa (lado dos componentes)



Lado cobreado (trilhas)
Imagem espelhada



Lado dos componentes
(máscara)
Imagem normal



Lado dos componentes
(referências)

Firmware da plataforma PIE

Para que a plataforma PIE opere adequadamente, é necessário que um *firmware* específico (programa de controle) seja gravado no microcontrolador PIC18F2550. Você pode baixar este *firmware* através do link http://vichinsky.com.br/inters/NEW_HID_INTERS.ZIP.

Para gravar o *firmware* você deve utilizar um gravador adequado. Sugerimos o gravador/programador PICKIT2, cujo manual pode ser obtido no site do fabricante (*Microchip Technology Inc.*): <http://ww1.microchip.com/downloads/en/devicedoc/51553e.pdf>.



5. Comandos INTER-S para operações com a plataforma PIE

Nesta seção são apresentados os comandos do Inter-S direcionados exclusivamente às operações com a Plataforma Experimental. Informações sobre esses comandos também podem ser encontradas no módulo de "Ajuda" do próprio *software*.

5.1. Estado_USB

Fornece o estado da porta USB, indicando se a Plataforma Inter-S Experimental (PIE) está conectada ou não.

Sintaxe: Var = Estado_USB

A variável "Var" deve ser declarada no início do programa como numérica.

Exemplo: X = Estado_USB

No exemplo acima, a variável numérica X receberá o valor que indica o estado da porta USB (se o dispositivo PIE estiver conectado, o valor de X será 1, caso contrário será 0)

Programa exemplo:

```
Rotina
  Declare U Numérico
  [VERIFICA]
  Limpa
  Escreva "Localizando dispositivo USB"
  U=ESTADO_USB
  Se U = 0 Então
    Escreva "Dispositivo USB não conectado."
    Escreva "Conecte a plataforma experimental e tecle algo."
    Pausa
    VaPara VERIFICA
  FimSe
  Escreva "Dispositivo USB conectado."
  ...
FimSubRotina
```

No exemplo acima, o algoritmo entrará em uma estrutura de repetição até que a plataforma PIE seja conectada ao computador.


5.2. Mostra_Placa

Esta instrução mostra uma imagem virtual da Plataforma Inter-S Experimental (PIE), na qual é possível observar o estado e o comportamento da mesma em tempo real, durante a execução do algoritmo. A exibição virtual da Plataforma Experimental independe de sua conexão física, sendo assim, é possível utilizar este recurso para realizar simulações que não requerem leitura das entradas.

Sintaxe: Mostra_Placa



A figura abaixo ilustra o processo de execução de um algoritmo que utiliza o comando "Mostra_Placa". Observe que a imagem virtual da Plataforma Experimental é exibida ao lado da janela de execução.



5.3. Ative digitais / analógicas

Ativa as entradas digitais ou analógicas.

Sintaxe: `Ative [digitais / analógicas]`

Exemplos: `Ative digitais` (Ativa as entradas digitais permitindo a leitura de seus valores)
`Ative analógicas` (Ativa as entradas analógicas permitindo a leitura de seus valores)

Observação: Por padrão, as entradas analógicas e digitais são ativadas na iniciação do algoritmo. O uso da instrução "Ative" é recomendado nos casos em que se deseja obter uma maior velocidade ao acesso às entradas. Para exemplificar, se utilizarmos a instrução "Ative digitais", o Inter-S fará apenas a leitura das entradas digitais em ciclos de 50 milissegundos, permitindo maior velocidade na atualização dos valores gerados pela plataforma PIE. Da mesma forma, se utilizarmos a instrução "Ative analógicas", o Inter-S fará apenas a leitura das analógicas em ciclos de 50 milissegundos.

4.4. Desative digitais / analógicas

Desativa as entradas digitais ou analógicas.

Sintaxe: `Desative [digitais / analógicas]`

Exemplos: `Desative digitais` (Desativa a leitura das entradas digitais)
`Desative analógicas` (Desativa a leitura das entradas analógicas)

5.5. Liga_Saída

Esta instrução envia um sinal de nível lógico alto para uma determinada saída digital da Plataforma Experimental PIE.

Sintaxe: `Liga_Saída n`

Onde n é o número da saída que será ligada, podendo ser de 0 a 7.



Exemplos: `Liga_Saída 0` (coloca a saída 0 em nível lógico alto - saída 0 ligada)

`s=3`

`Liga_Saída s` (coloca a saída 3 em nível lógico alto - saída 3 ligada)

5.6. Desliga_Saída

Esta instrução envia um sinal de nível lógico baixo para uma determinada saída digital da Plataforma Experimental.

Sintaxe: `Desliga_Saída n`

Onde n é o número da saída que será desligada, podendo ser de 0 a 7.

Exemplos: `Desliga_Saída 0` (coloca a saída 0 em nível lógico baixo - saída 0 desligada)

`s=5`

`Desliga_Saída s` (coloca a saída 5 em nível lógico baixo - saída 5 desligada)

5.7. Servo_Motor_n / LServo_Motor_n

Permite o controle de um servo-motor conectado à Plataforma Experimental. É possível controlar até 4 servos de 180° em uma mesma aplicação na plataforma PIE.

Sintaxe: `Servo_Motor_n <ang>` ou `LServo_Motor_n <ang>`

Onde n é o número do servo-motor, que pode ser 1, 2, 3 ou 4. O parâmetro <ang> corresponde ao ângulo para posicionamento do servo-motor indicado. A instrução `LServo_Motor_n` (iniciada com a letra L) promove o deslocamento lento, ao passo que a instrução `Servo_Motor_n` promove o deslocamento padrão (rápido).

Exemplos: `Servo_Motor_1 90` (Move o servo 1 para a posição 90°)

`a=45`

`Servo_Motor_2 a` (Move o servo 2 para a posição 45°)

5.8. Analógica

Faz a leitura de uma determinada entrada analógica da Plataforma Inter-S Experimental.

Sintaxe: `var = Analógica n`

Onde "var" é uma variável numérica declarada no início do programa e "n" é o número da entrada analógica (que pode ser 1, 2 ou 3).

Exemplos: `x = Analógica 1` (Atribui a x o valor existente na entrada analógica 1)

`z = Analógica 2` (Atribui a z o valor existente na entrada analógica 2)

Observação: Os valores possíveis de uma entrada analógica estão no intervalo de 0 a 255.



5.9. Digital

Faz a leitura de uma determinada entrada digital da Plataforma Inter-S Experimental (PIE).

Sintaxe: `var = Digital n`

Onde "var" é uma variável numérica declarada no início do programa e "n" é o número da entrada digital (que pode ser 1, 2, 3 ou 4).

Exemplos: `x = Digital 1` (Atribui a x o valor existente na entrada digital 1)
`z = Digital 2` (Atribui a z o valor existente na entrada digital 2)

Observação: Os valores possíveis de uma entrada digital são 0 (zero) e 1 (um). Na Plataforma Inter-S Experimental existem 4 botões táteis, dispostos em montagem do tipo *pull-up*, conectados aos terminais das entradas digitais. Esses botões podem ser utilizados em testes como entradas digitais 1, 2, 3 e 4, respectivamente, desde que ativados por meio do *jumper* J2 existente na plataforma. Quando um botão é pressionado, o valor do terminal associado a ele será 0 (zero), por outro lado, enquanto o botão estiver "solto", o valor do terminal associado a ele será 1 (um).

5.10. P_Saídas

Envia um valor entre 0 e 255 para ativação das saídas digitais da plataforma PIE.

Sintaxe: `P_Saídas n`

Onde n é um valor (constante numérica ou variável) entre 0 e 255.

Exemplos:

`P_Saídas 0` (envia o valor 0 para as saídas - desativa todas elas)

`P_Saídas 255` (envia o valor 255 para as saídas - ativa todas elas)

`x=128`

`P_Saídas x` (envia o valor 128 para as saídas - ativa apenas a saída 7)

Observação: Para entender o comportamento da função P_Saídas, é necessário converter o valor do parâmetro para a notação binária, de forma a obter 8 dígitos binários. Os dígitos, da direita para a esquerda, estão associados às saídas de 0 a 7, respectivamente. Se o dígito de uma determinada posição for 0 (zero), a saída associada a ele será desligada. Por outro lado, se o dígito for 1 (um), a saída associada a ele será ligada. Exemplo:

`P_Saídas 170`

O valor 170 do exemplo acima, convertido para notação binária, corresponde a 10101010. Lendo esse número binário da direita para a esquerda, podemos concluir que as saídas 0, 2, 4 e 6 serão desligadas, e as saídas 1, 3, 5 e 7 serão ligadas:

1 0 1 0 1 0 1 0 -> Número 170 convertido para binário

7 6 5 4 3 2 1 0 -> Número das saídas associadas às posições dos dígitos